Dans un premier temps, on désire intégrer numériquement une application f continue sur $[a,b]$; c'est à dire calculer une valeur approchée de $\int_a^b f(t)\,\mathrm{d}t$, appelée u_n, par la méthode des trapèzes, en découpant l'intervalle $[a,b]$ en n intervalles de même longueur.

Par la suite, on cherchera à améliorer le procédé.

1) L'aire dun trapèze est : $\dfrac{b-a}{n} \times \dfrac{f\left(a+k\,\frac{b-a}{n}\right)+f\left(a+(k+1)\,\frac{b-a}{n}\right)}{2}$, et il faut sommer ces aires de $k=0$ à $k=n-1$.
Ce qui donne :

$$
\begin{aligned}
u_n &= \frac{b-a}{2n} \times \sum_{k=0}^{n-1} \left(f\left(a+k\,\frac{b-a}{n}\right) + f\left(a+(k+1)\,\frac{b-a}{n}\right) \right) \\
&= \frac{b-a}{2n} \times \left(\sum_{k=0}^{n-1} f\left(a+k\,\frac{b-a}{n}\right) + \sum_{k=0}^{n-1} f\left(a+(k+1)\,\frac{b-a}{n}\right) \right) \\
&= \frac{b-a}{2n} \times \left(\sum_{k=0}^{n-1} f\left(a+k\,\frac{b-a}{n}\right) + \sum_{k=1}^{n} f\left(a+k\,\frac{b-a}{n}\right) \right) \\
&= \frac{b-a}{2n} \times \left(f(a) + \sum_{k=1}^{n-1} f\left(a+k\,\frac{b-a}{n}\right) + f(b) + \sum_{k=1}^{n-1} f\left(a+k\,\frac{b-a}{n}\right) \right) \\
&= \frac{b-a}{n} \left(\frac{f(a)+f(b)}{2} + \sum_{k=1}^{n-1} f\left(a+k\,\frac{b-a}{n}\right) \right)
\end{aligned}
$$

2) **a)** On écrit très facilement `Trapeze1` :

```
> Trapeze1:=proc(f,a,b,n)
> local a0,b0,d,u,k;
> a0:=evalf(a);b0:=evalf(b);d:=(b0-a0)/n;
> u:=d*((f(a0)+f(b0))/2+sum(f(a0+k*d),k=1..n-1));
> u
> end;
```

Trapeze1 := **proc**(f, a, b, n)
local $a0, b0, d, u, k$;
$a0 := \mathrm{evalf}(a)$;
$b0 := \mathrm{evalf}(b)$;
$d := (b0 - a0)/n$;
$u := d * (1/2 * f(a0) + 1/2 * f(b0) + \mathrm{sum}(f(a0 + k * d), k = 1..n-1))$;
u
end proc

b) On écrit facilement `Trapeze2`:

```
> Trapeze2:=proc(f,a,b,n)
> local a0,b0,d,u,k;
> a0:=evalf(a);b0:=evalf(b);d:=(b0-a0)/n;
> u:=d*((f(a0)+f(b0))/2);
> for k from 1 to n-1 do
> u:=u+d*f(a0+k*d)
> od;
> u
> end;
```

$$
\begin{array}{l}
\mathit{Trapeze2} := \mathbf{proc}(f, a, b, n)\\
\mathbf{local}\, a0, b0, d, u, k;\\
\quad a0 := \mathrm{evalf}(a)\,;\\
\quad b0 := \mathrm{evalf}(b)\,;\\
\quad d := (b0 - a0)/n\,;\\
\quad u := d * (1/2 * f(a0) + 1/2 * f(b0))\,;\\
\quad \mathbf{for}\, k\, \mathbf{to}\, n-1\, \mathbf{do}\ \ u := u + d * f(a0 + k * d)\ \mathbf{end\ do}\,;\\
\quad u\\
\mathbf{end\ proc}
\end{array}
$$

3) On écrit l'égalité aux rangs n et $2n$:

$$
\begin{aligned}
\int_a^b f(t)\,\mathrm{d}t &= u_n + \frac{K}{n^2} + \frac{L}{n^3} + \mathrm{o}\left(\frac{1}{n^3}\right)\\
\int_a^b f(t)\,\mathrm{d}t &= u_{2n} + \frac{K}{4n^2} + \frac{L}{8n^3} + \mathrm{o}\left(\frac{1}{n^3}\right)
\end{aligned}
$$

Quatre fois la seconde moins la première, le tout divisé par trois, élimine les $\dfrac{1}{n^2}$ dans $\displaystyle\int_a^b f(t)\,\mathrm{d}t$ et donne :

$$
\int_a^b f(t)\,\mathrm{d}t = \frac{4u_{2n} - u_n}{3} - \frac{L}{6n^3} + \mathrm{o}\left(\frac{1}{n^3}\right)
$$

On a donc $v_n = \dfrac{4u_{2n} - u_n}{3}$ et $L' = -\dfrac{L}{6}$.

4) On travaille d'abord u_{2n} :

$$
\begin{aligned}
u_{2n} &= \frac{b-a}{2n}\left(\frac{f(a)+f(b)}{2} + \sum_{k=1}^{2n-1} f\left(a + k\,\frac{b-a}{2n}\right)\right)\\
&= \frac{b-a}{2n}\left(\frac{f(a)+f(b)}{2} + \underbrace{\sum_{k=1}^{2n-1} f\left(a + k\,\frac{b-a}{2n}\right)}_{\text{avec } k \text{ pair}} + \underbrace{\sum_{k=1}^{2n-1} f\left(a + k\,\frac{b-a}{2n}\right)}_{\text{avec } k \text{ impair}}\right)\\
&= \frac{b-a}{2n}\left(\frac{f(a)+f(b)}{2} + \sum_{k'=1}^{n-1} f\left(a + k'\,\frac{b-a}{n}\right) + \sum_{k'=0}^{n-1} f\left(a + (2k'+1)\,\frac{b-a}{2n}\right)\right)\\
&= \frac{b-a}{2n}\left(\frac{f(a)+f(b)}{2} + \sum_{k=1}^{n-1} f\left(a + k\,\frac{b-a}{n}\right) + \sum_{k=0}^{n-1} f\left(a + \frac{2k+1}{2}\,\frac{b-a}{n}\right)\right)
\end{aligned}
$$

On calcule maintenant v_n :

$$
\begin{aligned}
v_n &= 2\,\frac{b-a}{3n}\left(\frac{f(a)+f(b)}{2} + \sum_{k=1}^{n-1} f\left(a + k\,\frac{b-a}{n}\right) + \sum_{k=0}^{n-1} f\left(a + \frac{2k+1}{2}\,\frac{b-a}{n}\right)\right)\\
&\quad - \frac{b-a}{3n}\left(\frac{f(a)+f(b)}{2} + \sum_{k=1}^{n-1} f\left(a + k\,\frac{b-a}{n}\right)\right)\\
&= \frac{b-a}{6n}\left(f(a) + f(b) + 2\sum_{k=1}^{n-1} f\left(a + k\,\frac{b-a}{n}\right) + 4\sum_{k=0}^{n-1} f\left(a + \frac{2k+1}{2}\,\frac{b-a}{n}\right)\right)
\end{aligned}
$$

5) On écrit très facilement `Integrale1`:

```
> Integrale1:=proc(f,a,b,n)
> local a0,b0,d,u,k;
> a0:=evalf(a);b0:=evalf(b);d:=(b0-a0)/n;
> u:=d/6*(f(a0)+f(b0)+2*sum(f(a0+k*d),k=1..n-1)+4*sum(f(a0+(2*k+1)*d/2),
> k=0..n-1));
> u
> end;
```

$$\begin{array}{l} \mathit{Integrale1} := \textbf{proc}(f,a,b,n) \\ \textbf{local}\; a0, b0, d, u, k; \\ \quad a0 := \mathrm{evalf}(a)\,; \\ \quad b0 := \mathrm{evalf}(b)\,; \\ \quad d := (b0 - a0)/n\,; \\ \quad u := 1/6 * d * (f(a0) + f(b0) + 2 * \mathrm{sum}(f(a0 + k * d), k = 1..n-1) \\ \qquad + 4 * \mathrm{sum}(f(a0 + 1/2 * (2 * k + 1) * d), k = 0..n-1)); \\ \quad u \\ \textbf{end proc} \end{array}$$

et on écrit facilement `Integrale2`:

```
> Integrale2:=proc(f,a,b,n)
> local a0,b0,d,u,k;
> a0:=evalf(a);b0:=evalf(b);d:=(b0-a0)/n;
> u:=d/6*(f(a0)+f(b0));
> for k from 1 to n-1 do
> u:=u+d/3*f(a0+k*d)
> od;
> for k from 0 to n-1 do
> u:=u+2*d/3*f(a0+(2*k+1)*d/2)
> od;
> u
> end;
```

$$\begin{array}{l} \mathit{Integrale2} := \textbf{proc}(f,a,b,n) \\ \textbf{local}\; a0, b0, d, u, k; \\ \quad a0 := \mathrm{evalf}(a)\,; \\ \quad b0 := \mathrm{evalf}(b)\,; \\ \quad d := (b0 - a0)/n\,; \\ \quad u := 1/6 * d * (f(a0) + f(b0))\,; \\ \quad \textbf{for}\; k\; \textbf{to}\; n-1\; \textbf{do}\;\; u := u + 1/3 * d * f(a0 + k * d)\;\; \textbf{end do}\,; \\ \quad \textbf{for}\; k\; \textbf{from}\; 0\; \textbf{to}\; n-1\; \textbf{do}\; u := u + 2/3 * d * f(a0 + 1/2 * (2 * k + 1) * d)\; \textbf{end do}\,; \\ \quad u \\ \textbf{end proc} \end{array}$$

6) On découpe l'intervalle $[0,\pi/2]$ en 10 pour caluler une valeur approchée de l'intégrale du `sinus`.
Ce qui fait travailler avec $n = 10$ pour `Trapeze1` et $n = 5$ pour `Integrale1`:

```
> Trapeze1(sin,0,Pi/2,10);
```

$$.9979429866$$

```
> Integrale1(sin,0,Pi/2,5);
```

$$1.000003393$$

L'accroissement de précision du calcul, le résultat exact est 1, est ici spectaculaire. Surtout quand on pense qu'on utilise exactement les mêmes valeurs de la fonction aux mêmes points...

7) On montre que pour une fonction polynomiale du second degré g :

$$\int_\alpha^\beta g(t)\,\mathrm{d}t = (\beta-\alpha)\,\frac{g(\alpha)+g(\beta)+4\,g\left(\frac{\alpha+\beta}{2}\right)}{6}$$

Pour cela, on pose $g(t) = at^2+bt+c$, on calcule l'intégrale puis $(\beta-\alpha)\,\dfrac{g(\alpha)+g(\beta)+4\,g\left(\frac{\alpha+\beta}{2}\right)}{6}$ et on compare.

$$\begin{aligned}\int_\alpha^\beta at^2+bt+c\,\mathrm{d}t &= a\,\frac{\beta^3-\alpha^3}{3}+b\,\frac{\beta^2-\alpha^2}{2}+c\,(\beta-\alpha)\\ &= \frac{\beta-\alpha}{6}\left(2a\left(\beta^2+\beta\alpha+\alpha^2\right)+3b\,(\beta+\alpha)+6c\right)\end{aligned}$$

$$\begin{aligned}g(\alpha)+g(\beta)+4\,g\left(\frac{\alpha+\beta}{2}\right) &= a\,\left(\alpha^2+\beta^2+4\,\left(\frac{\alpha+\beta}{2}\right)^2\right)+b\,\left(\alpha+\beta+4\,\left(\frac{\alpha+\beta}{2}\right)\right)+6c\\ &= \left(2a\left(\beta^2+\beta\alpha+\alpha^2\right)+3b\,(\beta+\alpha)+6c\right)\end{aligned}$$

On a bien l'égalité demandée.

Maintenant, il faut calculer

$$\begin{aligned}&\frac{b-a}{6n}\times\sum_{k=0}^{n-1}\left(f\left(a+k\,\frac{b-a}{n}\right)+f\left(a+(k+1)\,\frac{b-a}{n}\right)+4\,f\left(a+\frac{2k+1}{2}\,\frac{b-a}{n}\right)\right)\\ &\quad=\frac{b-a}{6n}\left(f(a)+f(b)+2\sum_{k=1}^{n-1}f\left(a+k\,\frac{b-a}{n}\right)+4\sum_{k=0}^{n-1}f\left(a+\frac{2k+1}{2}\,\frac{b-a}{n}\right)\right)\end{aligned}$$

en utilisant la même technique qu'à la première question.

Cela revient bien à changer, dans le calcul de u_n, le segment de droite du trapèze par un arc de parabole coincidant avec le graphe de f aux extrémités et au milieu du segment.